;     Езици за функционално и логическо програмиране.

;     Упражнение N11

;Задача 1. Нека [TREE] е списък, задаващ някакво дърво. Да се

;избере подходящо представяне на дървото, зададено чрез [TREE], и

;да се дефинират следните функции:

;а) (succ NODE) - връща списък от преките наследници на върха [NODE] в

;дървото [TREE];

;б) (leafp NODE) - връща #Т, ако върхът [NODE] е лист в дървото [TREE], и

;#F в противен случай;

;в) (list-leaves NODE) - връща списък от листата на дървото [TREE],

;които са наследници в дървото на възела [NODE];

;г) (paths NODE) - връща списък от пътищата в дървото [TREE] от

;възела [NODE] до всички листа на дървото, които са наследници на [NODE].

;д) (root) - намира корена на дървото [TREE];

;е) (leaves) - връща списък от листата на дървото [TREE];

;ж) (height) - намира височината на дървото [TREE] (максималната

;дължина на път от корена до лист от дървото);

;з) (commonpred a b) - намира най-близкия общ предшественик на двата

;възела а и b, така, че поддървото с корен този възел съдържа и двата

;възела.

;Пример: (list-leaves 'C) -> (F I J H)

;(list-leaves 'A) -> (D E F I J H)

;(paths 'C) -> ((C F) (C G I) (C G J) (C H))

;(paths 'A) -> ((A B D ) (A B E) (A C F) (A C G I) (A C G J) (A C H))

;Решение: - чрез асоциативен списък.

;Представяне:

;TREE -> ((<възел 1> <наследник 1> <наследник 2> ...)

;
  (<възел 2> <наследник 1> <наследник 2> ...)

;
   ...)

;Дефиниции:

 (define tree '((a b c) (b d e) (c f g h) (d) (e) (f) (g i j)


       
 (h) (i) (j)))

(define (succ node)

  (cdr (assoc node tree)))

;Различни при двете представяния са само дефинициите на функцията

;succ. Всички останали функции използват тази функция и така
;техните дефиниции се оказват независими от представянето. По този

;начин беше разгледан добър пример за абстракция на данни.

 (define (leafp node)

  (null? (succ node)))

(define (list-leaves node)

  (if (leafp node)

      (list node)

      (apply append (map list-leaves (succ node)))))

(define (paths node)

;;; Връща (([node])), когато [node] е лист от дървото

   (if (leafp node)

       (list (list node))

       (map (lambda (x) (cons node x))

            (apply append

                   (map paths 

                        (succ node)))) )) 
(define (root)

  (define (root1 tree1)

    (cond ((null? tree1) ())


     ((not (parent (caar tree1) tree)) (caar tree1))


     (else (root1 (cdr tree1)))))

  (root1 tree))
;(parent node tree) - за възел node връща родителя му в tree или #F.

(define (parent node tree)

  (cond ((null? tree) #F)

 
   ((member node (succ (caar tree))) (caar tree))

   (else (parent node (cdr tree)))))

(define (leaves)     
  (list-leaves (root)))  

(define (height)         
  (apply max (map length (paths (root)))))

(define (commonpred a b)

  (cond ((eq? a b) a)

   ((pred? a b) a)


   ((pred? b a) b)


   (else (commonpred (parent a tree) b))))

;(pred? a b) - #Т, ако а е предшественик (родител) на b, или #F.

(define (pred? a b)

  (cond ((leafp a) #F)


   ((eq? (root) b) #F)


   ((eq? (parent b tree) a) #T)


   (else (pred? a (parent b tree)))))

